
AxonInterp
    

The thin wrapper for Axon inference.
It is a Deep Learning inference framework that can be used in the same way as my *Interp.
[ATTENSION: AxonInterp v0.1.0 is experimental release!]
Platform
I have confirmed it works in the following OS environment.
	WSL2/Ubuntu 20.04

Installation
You can install it by adding axon_interp to the mix.exs dependency list as follows:
def deps do
  [
    {:axon_interp, "~> 0.1.0"}
  ]
end
Basic Usage
You get the trained Axon model or Onnx model and save it in a directory that your application can read.
"your-app/priv" may be good choice.
$ cp your-trained-model.axon ./priv

Next, you will create a module that interfaces with the deep learning model. The module will need pre-processing and
post-processing in addition to inference processing, as in the example following. AxonInterp provides inference processing only.
You put use AxonInterp at the beginning of your module, specify the model's path as an optional argument and describe
model's inputs/outputs specification. In the inference, you will put data input to the model (AxonInterp.set_input_tensor/3),
inference execution (AxonInterp.invoke/1), and inference result retrieval (AxonInterp.get_output_tensor/2).
Currentrly inputs/outputs of the medel are binary data format, because of easy to share the module with other *Interp.
defmodule YourApp.YourModel do
  use AxonInterp,
    model: "priv/your-trained-model.axon",
    inputs: [f32: {1, 3, 224, 224}],
    outputs: [f32: {1, 1000}]

  def predict(data) do
    # preprocess
    #  to convert the data to be inferred to the input format of the model.
    input_bin = convert-float32-binaries(data)

    # inference
    #  typical I/O data for Onnx models is a serialized 32-bit float tensor.
    output_bin = session()
      |> AxonInterp.set_input_tensor(0, input_bin)
      |> AxonInterp.invoke()
      |> AxonInterp.get_output_tensor(0)

    # postprocess
    #  add your post-processing here.
    #  you may need to reshape output_bin to tensor at first.
    tensor = output_bin
      |> Nx.from_binary(:f32)
      |> Nx.reshape({1000})

    * your-postprocessing *
    ...
  end
end
Demo
There is Object Detection: YOLOv4 example in the demo_yolov4 directory.
This demo artistically converts a photo of a frog 'flog.jpg' in the demo directory and saves it as 'candy.jpg'.
First, you download the trained DNN model "candy-9.onnx" from the following URL and place it in the demo directory.
	candy-9.onnx: https://github.com/onnx/models/raw/main/vision/style_transfer/fast_neural_style/model/candy-9.onnx

You can run the demo by following these steps.
$ cd demo
$ mix deps.get
$ mix run -e "DemoCandy.run"

You get "candy.jpg" in current directory.
Let's enjoy ;-)
License
AxonInterp is licensed under the Apache License Version 2.0.

Release 0.1.0(Dec 3 2022)
	Experimental version




  

    
Resnet18 image classification
    

File.cd!(__DIR__)
# for windows JP
# System.shell("chcp 65001")

Mix.install(
  [
    {:axon_interp, "~> 0.1.0"},
    {:cimg, "~> 0.1.14"},
    {:nx, "~> 0.4.0"},
    {:kino, "~> 0.7.0"}
  ],
  config: [
    nx: [default_defn_options: [compiler: EXLA]]
  ]
  # system_env: [{"NNINTERP", "Axon"}]
)
0.Original work
torchvision.models.resnet18 -  pre-trained model included in Pytorch.
	https://pytorch.org/vision/main/models/generated/torchvision.models.resnet18.html

Thanks a lot!!!

Implementation with AxonInterp in Elixir
1.Defining the inference module: ResNet18
	Model<br>
resnet18.axon: get from "https://github.com/shoz-f/axon_interp/releases/download/0.0.1/resnet18.axon" if not existed.

	Pre-processing<br>
Resize the input image to the size {@width, @height} and gaussian normalize.

	Post-processing<br>
Sort outputs and take first item.


defmodule Resnet18 do
  @width 224
  @height 224

  use AxonInterp,
    model: "./model/resnet18.onnx",
    url: "https://github.com/shoz-f/axon_interp/releases/download/0.0.1/resnet18.onnx",
    inputs: [f32: {1, 3, @height, @width}],
    outputs: [f32: {1, 1000}]

  @imagenet1000 (for item <- File.stream!("./model/imagenet1000.label") do
                   String.trim_trailing(item)
                 end)
                |> Enum.with_index(&{&2, &1})
                |> Enum.into(%{})

  def apply(img, top \\ 1) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.to_binary([{:gauss, {{123.7, 58.4}, {116.3, 57.1}, {103.5, 57.4}}}, :nchw])

    # prediction
    output0 =
      session()
      |> AxonInterp.set_input_tensor(0, input0, [:binary])
      |> AxonInterp.invoke()
      |> AxonInterp.get_output_tensor(0)
      |> Nx.squeeze()

    # postprocess
    then(Nx.exp(output0), fn exp -> Nx.divide(exp, Nx.sum(exp)) end)
    |> Nx.argsort(direction: :desc)
    |> Nx.slice([0], [top])
    |> Nx.to_flat_list()
    |> Enum.map(&@imagenet1000[&1])
  end
end
Launch Resnet18.
# AxonInterp.stop(Resnet18)
Resnet18.start_link([])
Display the properties of the Resnet18 model.
AxonInterp.info(Resnet18)
2.Let's try it
Load a photo and apply Resnet18 to it.
img = CImg.load("lion.jpg")
Kino.render(CImg.display_kino(img, :jpeg))

Resnet18.apply(img, 3)
3.Appendix - How to convert Pytorch model to Axon
STEP1.
Create Onnx model from torchvision.models.resnet18.
mk_resnet18_onnx.py
import os
import torch
import torchvision.models as models

model = models.resnet18(weights='DEFAULT')
model.eval()

dummy_input = torch.randn(1, 3, 224, 224)

os.makedirs("data", exist_ok=True)
torch.onnx.export(model,
   dummy_input,
   "data/resnet18.onnx",
   verbose=False,
   input_names=["input.0"],
   output_names=["output.0"],
   export_params=True
   )
STEP2.
Convert Onnx model to Axon's graph and parameter.
mk_resnet18_axon.exs
Mix.install([
  {:axon_onnx, "~> 0.3.0"}
])

AxonOnnx.import("data/resnet18.onnx")
|> then(fn {model, params} -> Axon.serialize(model, params) end)
|> (&File.write("data/resnet18.axon", &1)).()

Object Detection: YOLOv4
    

File.cd!(__DIR__)
# for windows JP
# System.shell("chcp 65001")

Mix.install(
  [
    {:axon_interp, "~> 0.1.0"},
    {:cimg, "~> 0.1.14"},
    {:postdnn, "~> 0.1.5"},
    {:kino, "~> 0.7.0"}
  ],
  config: [
    nx: [default_defn_options: [compiler: EXLA]]
  ]
  # system_env: [{"NNINTERP", "Axon"}]
)
0.Original work
Pytorch-YOLOv4
	https://github.com/Tianxiaomo/pytorch-YOLOv4

This note usee the pretraind model converted from above project ;-)
Thanks a lot!!!

Implementation with AxonInterp in Elixir
1.Defining the inference module: YOLOv4
	Model<br>
yolov4_1_3_608_608_static.onnx

	Pre-processing<br>
Resize the input image to the size {608,608}, normalize to {0.0,1.0} and transpose NCHW.

	Post-processing<br>
Filtering Boxes(output[0]) and Scores(output[1]) with Multi-class Non Maximum Suppression.


defmodule YOLOv4 do
  @moduledoc """
  Original work:
    Pytorch-YOLOv4 - https://github.com/Tianxiaomo/pytorch-YOLOv4
  """

  use AxonInterp,
    model: "./model/yolov4_1_3_608_608_static.onnx",
    url:
      "https://github.com/shoz-f/axon_interp/releases/download/0.0.1/yolov4_1_3_608_608_static.onnx",
    inputs: [f32: {1, 3, 608, 608}],
    outputs: [f32: {1, 22743, 1, 4}, f32: {1, 22743, 80}]

  def apply(img) do
    # preprocess
    input0 =
      img
      |> CImg.resize({608, 608})
      |> CImg.to_binary([{:range, {0.0, 1.0}}, :nchw])

    # prediction
    outputs =
      session()
      |> AxonInterp.set_input_tensor(0, input0, [:binary])
      |> AxonInterp.invoke()

    # postprocess
    boxes = AxonInterp.get_output_tensor(outputs, 0, [:binary])
    scores = AxonInterp.get_output_tensor(outputs, 1, [:binary])

    PostDNN.non_max_suppression_multi_class(
      {22743, 80},
      boxes,
      scores,
      boxrepr: :corner,
      label: "./model/coco.label"
    )
  end
end
Launch YOLOv4.
# AxonInterp.stop(Resnet18)
YOLOv4.start_link([])
Display the properties of the YOLOv4 model.
AxonInterp.info(YOLOv4)
2.Defining execution module LiveYOLOv4
defmodule LiveYOLOv4 do
  @palette CImg.Util.rand_palette("./model/coco.label")

  def run(path) do
    img = CImg.load(path)

    with {:ok, res} <- YOLOv4.apply(img) do
      IO.inspect(res)

      Enum.reduce(res, CImg.builder(img), &draw_item(&1, &2))
      |> CImg.display_kino(:jpeg)
    end
  end

  defp draw_item({name, boxes}, canvas) do
    color = @palette[name]

    Enum.reduce(boxes, canvas, fn [_score, x1, y1, x2, y2, _index], canvas ->
      [x1, y1, x2, y2] = PostDNN.clamp([x1, y1, x2, y2], {0.0, 1.0})

      CImg.fill_rect(canvas, x1, y1, x2, y2, color, 0.35)
    end)
  end
end
3.Let's try it
LiveYOLOv4.run("dog.jpg")
Appendix

It is a Deep Learning inference framework that can be used in the same way as my *Interp.

Execute the inference session. In session mode, data input/execution of

  Examples.


    output_bin0 = session()
      |> AxonInterp.set_input_tensor(0, input_bin0)
      |> AxonInterp.invoke()
      |> AxonInterp.get_output_tensor(0)

Put a flat binary to the input tensor on the interpreter.

  
  parameters

  
  Parameters


	sessin - session record
	index - index of input tensor in the model
	bin   - input data - flat binary, cf. serialized tensor
	opts	[:binary] - data is binary

Download and process data from url.

  
  parameters

  
  Parameters


	url - download site url
	func - function to process downloaded data

